

January, 1996

## Advisor Answers

Q: Is there an easy way of converting/mapping an Access .MDB file to individual .DBF files. I have tried the Remote View feature in Visual FoxPro and I can link tables or view them, but I can't copy them to .DBF tables.

–Michael Verry (via CompuServe)

A: Turning remote data into local data is a two-step process. First, you open a remote view of the data, which copies it into a cursor. Then, you need to copy from the cursor, which is a temporary table, into a permanent table.

If you've already defined the view, it looks like this:

```
USE MyRemoteView
COPY TO MyNewTable
```

For an Access .MDB, though, there's probably a lot more information than a single table. If you want to convert the entire database, it's probably easier to use SQL Pass-Through (SPT) than views. SPT offers functions that let you know what's in the database, so you can process it. Try something like this:

```
FUNCTION ConvRemo
* Convert remote data into local data

LOCAL nHandle,nSuccess, cQuery

* Substitute your data source in the next line
nHandle = SQLConnect("Access with ODBC 2.0")
IF nHandle>0 && Successfully connected
  * Get a list of tables in the remote database
  nSuccess = SQLTABLES(nHandle,"TABLE","Remote")

  IF nSuccess>0
    * Process each table
    SCAN
      * Open the table
      cQuery = "SELECT * FROM "+ALLTRIM(Table_Name)
      nSuccess = SQLEXEC(nHandle,cQuery)
      IF nSuccess > 0
        COPY TO ALLTRIM(Remote.Table_Name)
      ENDIF
    ENDSCAN
  ENDIF
ENDIF
```

Of course, you'll want to not just check for errors, but do something when they occur. You'll probably also want to check the table names for validity in FoxPro. (When I run this code against the Access Northwinds sample data, several of the table names get truncated.)

With a little more code, you can actually create a Visual FoxPro database and put the new tables in it. Take a look at the ADD TABLE command. And, if you can get the server

(Access, in your example) to hand you the information about rules and triggers and so forth, you can fill in that information, too. Look at DBSETPROP() for that kind of thing.

The version of ConvRemo shown here is on this month's Companion Resource Disk.

-Tamar